<commit_before>{ stdenv, fetchurl,
  boost, clasp, cmake, gringo, re2c
}:

let
  version = "1.9.1";
in

stdenv.mkDerivation rec {
  name = "aspcud-${version}";

  src = fetchurl {
    url = "mirror://sourceforge/project/potassco/aspcud/${version}/aspcud-${version}-source.tar.gz";
    sha256 = "09sqbshwrqz2fvlkz73mns5i3m70fh8mvwhz8450izy5lsligsg0";
  };

  buildInputs = [ boost clasp cmake gringo re2c ];

  buildPhase = ''
    cmake -DCMAKE_BUILD_TYPE=Release \
      -DGRINGO_LOC=${gringo}/bin/gringo \
      -DCLASP_LOC=${clasp}/bin/clasp \
      -DENCODING_LOC=$out/share/aspcud/specification.lp \
      .

    make
  '';

  installPhase = ''
    mkdir -p $out/bin
    cp bin/{aspcud,cudf2lp,lemon} $out/bin

    mkdir -p $out/share/aspcud
    cp ../share/aspcud/specification.lp $out/share/aspcud
  '';

  meta = with stdenv.lib; {
    description = "Solver for package problems in CUDF format using ASP";
    homepage = http://potasssco.sourceforge.net/;
    platforms = platforms.all;
    maintainers = [ maintainers.hakuch ];
    license = licenses.gpl3Plus;
  };
}<commit_msg>aspcud: 1.9.1 -> 1.9.4<commit_after>{ stdenv, fetchzip
, boost, clasp, cmake, gringo, re2c
}:

stdenv.mkDerivation rec {
  version = "1.9.4";
  name = "aspcud-${version}";

  src = fetchzip {
    url = "https://github.com/potassco/aspcud/archive/v${version}.tar.gz";
    sha256 = "0vrf7h7g99vw1mybqfrpxamsnf89p18czlzgjmxl1zkiwc7vjpzw";
  };

  buildInputs = [ boost clasp cmake gringo re2c ];

  cmakeFlags = [
    "-DCMAKE_BUILD_TYPE=Release"
    "-DASPCUD_GRINGO_PATH=${gringo}/bin/gringo"
    "-DASPCUD_CLASP_PATH=${clasp}/bin/clasp"
  ];

  meta = with stdenv.lib; {
    description = "Solver for package problems in CUDF format using ASP";
    homepage = "https://potassco.org/aspcud/";
    platforms = platforms.all;
    maintainers = [ maintainers.hakuch ];
    license = licenses.gpl3Plus;
  };
}
{'fraction_non_alphanumeric': 0.14648334214701217, 'fraction_numerical': 0.028027498677948175, 'mean_word_length': 3.251685393258427, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 4, 'https://': 2, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '5506271', 'n_tokens_mistral': 806, 'n_tokens_neox': 755, 'n_words': 122}